

〔講座〕

パソコンで見る流体力学(その2)

—翼まわりの流れ—



石綿良三*

前回⁽¹⁾は回転する円柱まわりのポテンシャル流れについて解説したが、これにある写像(座標変換)を施すことによって翼まわりの流れを得ることができる。今回はこのようにして得られる翼まわりのポテンシャル流れを取り上げる。

1. ジューコフスキー写像

複素平面内の点 $\zeta(=\xi+i\eta)$ を点 $z(=x+iy)$ に変換する次の写像をジューコフスキー写像という。

$$z = \zeta + \frac{c^2}{\zeta} \quad \cdots(1)$$

ここで、 c は定数である。これを用いると、図1(a)の ζ 平面内の中心が (ξ_0, η_0) で点 $(c, 0)$ を通る円は、 z 平面では図1(b)のような翼形に変換される。この翼形をジューコフスキー翼形と呼んでいる。式(1)に $\zeta = \xi + i\eta$ 、 $z = x + iy$ を代入し、実数部と虚数部に分解すると、次の関係が得られる。

$$\left. \begin{aligned} x &= \xi + \frac{c^2 \xi}{\xi^2 + \eta^2} \\ y &= \eta - \frac{c^2 \eta}{\xi^2 + \eta^2} \end{aligned} \right\} \quad \cdots(1')$$

これらの式の (ξ, η) に円周上の座標を代入していき、計算される点 (x, y) を結んでいけばジューコフスキー翼形が得られる。 ξ_0 と η_0 はそれぞれ任意に設定でき、後述するように ξ_0 は翼厚、 η_0 はそれに影響するパラメータとなる。なお、 ζ 平面内の円の半径 a は次式で得られる。

$$a = \sqrt{(c - \xi_0)^2 + \eta_0^2} \quad \cdots(2)$$



(a) ζ 平面(b) z 平面

図1 ジューコフスキー写像

2. ジューコフスキー翼形のプログラム

ξ_0 と η_0 をパラメータとして翼形を描くプログラムをプログラム4に示す。140～200行では文字や枠を描いている。230行の関数 FNX、FNY はそれぞれ式(1')に対応し、 (x, y) をさらにグラフィック画面上の座標(FNX, FNY)に変換するものである。250行では ξ_0 、 η_0 を変化させながらそれぞれの翼形を描いている。サブルーチンの280～330行は一つひとつの翼形を描くものでSOが ξ_0 、TOが η_0 に対応している。

* 神奈川工科大学
原稿受付日 平成2年11月2日

```

100 '----- JOUKOWSKI -----
110 '----- (main) -----
120 SCREEN 3 : CONSOLE 0,25,0,1 : CLS 3
130 CC=15 : P=6.2832
140 LOCATE 28,0 : PRINT"ジュコフスキー翼";
150 LINE(0,24)-(639,399),1,B : LINE(0,24)-(64,64),1
160 FOR I=0 TO 6 : Y=48*I+64 : LINE(0,Y)-(639,Y),1
170 LOCATE 2,3*I+5 : PRINT USING"###";-I*.05; : NEXT
180 FOR J=0 TO 5 : X=96*J+64 : LINE(X,24)-(X,399),1
190 LOCATE 12*J+13,2 : PRINT USING"###";J/10; : NEXT
200 LOCATE 5,2 : PRINT"eta"; : LOCATE 1,3 : PRINT"xi";
210 '----- FUNCTION -----
220 DEF FNR=S^2+T^2
230 DEF FNX=X0+S*S*CC^2/FNR : DEF FNY=Y0-T+T*CC^2/FNR
240 '----- AEROFOIL SECTION -----
250 FOR I=0 TO 6 : FOR J=0 TO 5 : GOSUB 280 : NEXT J,I
260 LOCATE 0,0 : END
270 '----- (subroutine) -----
280 S0=-CC*I*.05 : Y0=96*I*48 : T0=CC*J/10 : X0=112+J*96
290 A=SQR((CC-S0)^2+T0^2) : F0=ATN(T0/(CC-S0))
300 X1=X0+2*CC : Y1=Y0
310 FOR F=0 TO P STEP .1 : S=S0+A*COS(F-F0) : T=T0+A*SIN(F-F0)
320 X=FNX : Y=FNY : LINE(X1,Y1)-(X,Y),5 : X1=X : Y1=Y : NEXT
330 LINE(X1,Y1)-(X0+2*CC,Y0),5 : RETURN

```

プログラム4 ジュコフスキー翼形




図2 ジュコフスキー翼形

このプログラム4を実行すると図2が得られる。図中の「eta」は η_0/c 、「xi」は ξ_0/c に対応し、それぞれのジュコフスキー翼形を描いている。 ξ_0 は翼厚に、 η_0 はそれに影響を与えていることがわかる。

3. 翼まわりの流れ

ここではジュコフスキー翼まわりのポテンシャル流れについて解説する。図1(a)に示した ξ 平面内の円柱が迎え角 α (ξ 軸とのおなす角が α)の一樣流中で回転している場合、複素ポテンシャル W は次式で表される。

$$W = Ue^{-id}(\zeta - \zeta_0) + \frac{Ua^2 e^{ia}}{\zeta - \zeta_0} + i \frac{\Gamma}{2\pi} \log(\zeta - \zeta_0) \quad \cdots(3)$$

ここで、右辺第1、2項は迎え角 α の一様流中の円柱まわりの流れ、第3項は循環 Γ の回転を表し、 U は一様流の速度、 $\zeta_0 = \xi_0 + i\eta_0$ 、半径 a は式(2)によって求められる。 ζ 平面内の速度を求めるため、式(3)を ζ で微分すると次式が得られる。

$$\frac{dW}{d\zeta} = Ue^{-ia} - \frac{Ua^2 e^{ia}}{(\zeta - \zeta_0)^2} + i \frac{\Gamma}{2\pi} \cdot \frac{1}{\zeta - \zeta_0} \quad \cdots(4)$$

ζ 平面内の ξ 方向速度を v_ξ 、 η 方向速度を v_η とすると、式(4)の実数部が v_ξ 、虚数部が v_η となり、

$$v_\xi = U \cos \alpha - \frac{Ua^2 \{(s^2 - t^2) \cos \alpha + 2st \sin \alpha\}}{(s^2 + t^2)^2} + \frac{\Gamma}{2\pi} \cdot \frac{t}{s^2 + t^2} \quad \cdots(5)$$

$$v_\eta = U \sin \alpha + \frac{Ua^2 \{(s^2 - t^2) \sin \alpha - 2st \cos \alpha\}}{(s^2 + t^2)^2} - \frac{\Gamma}{2\pi} \cdot \frac{s}{s^2 + t^2} \quad \cdots(6)$$

ただし、 $s = \xi - \xi_0$ 、 $t = \eta - \eta_0$ である。

式(4)および(5)、(6)における循環 Γ の大きさは翼後縁のクッタの条件から決定できる。クッタの条件によれば、後縁($\xi = c$, $\eta = 0$) は一つのよどみ点となる。そこで式(6)に $v_\eta = 0$ を代入すると、次式のように循環を求めることができる。

$$\frac{\Gamma}{2\pi} = U \left\{ \sin \alpha + \frac{a^2 \left[(s^2 - t^2) \sin \alpha - 2st \cos \alpha \right]}{(s^2 + t^2)^2} \right\} \frac{s^2 + t^2}{s} \quad \cdots(7)$$

ただし、 $s = c - \xi_0$ 、 $t = -\eta_0$ である。

さらに z 平面内の x 方向速度 u 、 y 方向速度 v を求めるためには以下のように dW/dz を計算すればよい。

まず、式(1)を ζ で微分し、

$$\frac{dz}{d\zeta} = 1 - \frac{c^2}{\zeta^2} \quad \cdots(8)$$

dW/dz は次のように置き換えられる。

$$\frac{dW}{dz} = \left(\frac{dW}{d\zeta} \right) / \left(\frac{dz}{d\zeta} \right) \quad \cdots(9)$$

これに式(4)と(8)を代入し、実数部と虚数部に分けるとそれぞれ u と $-v$ になる。以上によって z 平面内の速度場が求められる。

4. 翼まわりの流れの動画表示

翼まわりの流れを動画表示するために前回⁽¹⁾と同様にパレットの切り換えによるトレーサ表示を用いた。その際、式(9)から z 平面内の速度 u と v を計算する方法もあるが、ここでは式(5)、(6)によって ζ 平面内におけるトレーサ粒子の位置(ξ , η)を計算し、それを式(1')によって z 平面内の座標に変換する方法を用いた。

プログラムリストをプログラム5に示す。180行と190行はそれぞれ式(5)と(6)、200行は式(1')のジューコフスキー写像を表している。300、310行では式(7)によって $\Gamma/2\pi$ を求めている。330~360行では翼形を描き380~390行ではトレーサの軌跡をルンゲ・クッタ法(450~520行)によって求めている。410~420行および540~580行は動画部分である。

プログラムの使用法は次の通りである。まず ξ_0/c 、 η_0/c 、迎え角 α 、および流線の本数を入力する。動画が始まると、「+」キーで加速、「-」キーで減速、スペース・キーで一時停止とその解除ができることは回

```

100 '----- FLOW AROUND A AEROFOIL -----
110 '----- (main) -----
120 SCREEN 3 : CONSOLE 0,25,0,1 : CLS 3
130 ON STOP GOSUB 600 : STOP ON
140 '----- PARAMETER -----
150 CC=50 : U=5 : X0=319 : Y0=199 : C=1 : SP=200 : P=6.2832
160 '----- FUNCTION -----
170 DEF FNR=S^2+T^2
180 DEF FNU=U*CA-U*A^2*(CA*(S^2-T^2)+2*SA*S*T)/FNR^2+G*T/FNR
190 DEF FNV=U*SA+U*A^2*(SA*(S^2-T^2)-2*CA*S*T)/FNR^2-G*S/FNR
200 DEF FNX=X0+S*S*CC^2/FNR : DEF FNY=Y0-T*T*CC^2/FNR

```

```

210 '----- INPUT DATA -----
220 LOCATE 25,24 : PRINT"ジュコフスキー翼まわりの流れ";
230 LOCATE 0,0 : INPUT"xi/c ";S0 : S0=CC*S0
240 INPUT"eta/c";T0 : T0=CC*T0
250 INPUT"迎え角(*)";AL : AL=AL*P/360
260 INPUT"流線数";N : DIM S(N),T(N),E(N)
270 SA=SIN(AL) : CA=COS(AL)
280 FOR I=1 TO N : S(I)=-X0+20 : T(I)=2*Y0/N*I-Y0 : NEXT
290 A=SQR((CC-S0)^2+T0^2) : F0=ATN(T0/(CC-S0))
300 S=CC-S0 : T=-T0
310 G=U*(SA+A^2*(SA*(S^2-T^2)-2*CA*S*T)/FNR^2)/S*FNR
320 '----- AEROFOIL SECTION -----
330 X1=X0+2*CC : Y1=Y0
340 FOR F=0 TO P STEP .1 : S=S0+A*COS(F-F0) : T=T0+A*SIN(F-F0)
350 X=FNX : Y=FN Y : LINE(X1,Y1)-(X,Y),7 : X1=X : Y1=Y : NEXT
360 LINE(X1,Y1)-(X0+2*CC,Y0),7 : PAINT(X0-2*CC+2,Y0),7,7
370 '----- DRAWING -----
380 EE=1 : FOR I=1 TO N : IF E(I)=0 THEN GOSUB 450
390 NEXT : C=(C MOD 6)+1 : IF EE=0 THEN GOTO 380
400 '----- ANIMATION -----
410 COLOR=(C1,0) : COLOR=(C,6) : C1=C : C=(C MOD 6)+1
420 GOSUB 540 : FOR J=0 TO SP : NEXT : GOTO 410
430 '----- (subroutine) -----
440 '----- RUNGE-KUTTA -----
450 S1=S(I)-S0 : T1=T(I)-T0 : S=S1 : T=T1 : K1=FNU : L1=FN V
460 S=S1+K1/2 : T=T1+L1/2 : K2=FNU : L2=FN V
470 S=S1+K2/2 : T=T1+L2/2 : K3=FNU : L3=FN V
480 S=S1+K3 : T=T1+L3 : K4=FNU : L4=FN V
490 S=S(I) : T=T(I) : X1=FN X : Y1=FN Y
500 S(I)=S(I)+(K1+2*K2+2*K3+K4)/6 : T(I)=T(I)+(L1+2*L2+2*L3+L4)/6
510 S=S(I) : T=T(I) : X=FN X : Y=FN Y : IF X>639 THEN E(I)=1
520 LINE(X1,Y1)-(X,Y),C : EE=EE*E(I):RETURN
530 '----- SPEED CONTROL -----
540 S$=INKEY$ : IF S$="" THEN RETURN
550 IF S$="+" THEN SP=SP/1.05
560 IF S$="-" THEN SP=SP*1.05
570 IF S$<>" " THEN RETURN
580 IF INKEY$="" THEN RETURN ELSE 580
590 '----- STOP ROUTINE -----
600 COLOR : END

```

プログラム5 ジュコフスキー翼まわりの流れ




図3 ジュコフスキー翼まわりの流れの動画表示

転円柱のプログラム1⁽¹⁾と同じである。また150行の「CC」は c の値であり、これを書き換えることによって翼の大きさを変えることもできる。

プログラム5の実行例を図3、4に示す。図3は動

画のある瞬間の画面を示しており、(a)は $\xi_0/c = -0.1$ 、 $\eta_0/c = 0.1$ 、(b)は $\xi_0/c = -0.15$ 、 $\eta_0/c = 0.4$ の翼形まわりの流れである。動画を見ると翼上面に比べて下面の速度が遅く、したがって下面の圧力が大きくなり、



(a) $\xi_0/c = -0.1$, $\eta_0/c = 0.1$, $\alpha = 5^\circ$, 流線数=30(b) $\xi_0/c = -0.2$, $\eta_0/c = 0.2$, $\alpha = 5^\circ$, 流線数=30
($c=100$)

図4 ジューコフスキー翼まわりの流線

揚力が発生することがわかる。

図4はプリンタに画面のハード・コピーを行った例であり、流線が出力されている。翼によって流れが下向きに曲げられていることから、運動量理論によって揚力が発生することを説明できる。なお、図4(b)では $CC=100$ とおいて翼を大きく描いている。

プログラム5は流線数を大きくすると実行時間がかかるので、一度描いた画面をディスクにセーブしておく方法もある。プログラム5を実行した後、グラフィック画面をそのままの状態データ用ディスクをディスク・ドライブ2にセットし、プログラム2⁽¹⁾でセーブを行い、その後必要な時にプログラム3⁽¹⁾で動画表示を行えばよい。

5. まとめ

ジューコフスキー写像によって得られるジューコフスキー翼形を描くプログラム4では、パラメータのとり方によって翼形がどのように変化するかが明らかとなった。そのまわりのポテンシャル流れを動画表示するプログラム5では、流れの様子や揚力発生機構が観察できる。

〈参考文献〉

- (1) 石綿、パソコンで見る流体力学(その1)、ターボ機械、18-12(平成2-12)、718.